HEWLETT-PACKARD

Plotter ROM

OWNER'S MANUAL

HP-86/87

HEWLETT PACKARD

Plotter ROM
Owner's Manual

HP-86/87

June 1982

Reorder Number
00087-90109

Printed in U.S.A.

Blank Page

Contents

Blank Page

Section 1

Getting Started

Introduction

The Plotter ROM enables you to interface your computer with any of the Hewlett-Packard plotters which use HP-GL (Hewlett-Packard Graphics Language) instructions. The ROM extends all the CRT graphics capabilities to external plotters except byte plotting (BPLOT) and byte reading (BREAD) operations. The Plotter ROM also enables you to copy the CRT alpha and graphics displays, dot by dot, to a printer.

Hewlett-Packard plotters extend your graphics capabilities by offering the following features:

- *A larger plotting area:* The available plotting area for Hewlett-Packard plotters is considerably larger than your computer's graphics display. For example, the HP 7470A Plotter offers a 272 by 191 millimeter maximum plotting area compared to the 171 by 75 millimeter HP-87 *graph-all* mode display size. The maximum plotting area is device-dependent. Refer to the documentation for your plotter to find out the maximum plotting area dimensions.
- *Pen color/width selection:* External pen-plotters enable you to plot in different colors and line widths. Some plotters change pens automatically, while others require manual pen changes.
- *Digitizing:* The term *digitize* refers to the process of converting data into its digital form. For example, a surveyor who converts the graphical information on a topographic map to its digital form—latitude, longitude, and elevation, is digitizing graphics. Maps, photographs, schematic drawings, charts, and many other types of graphics can be digitized using an external plotter. Digitized graphics can be stored as numeric data and plotted on the CRT or an external plotter.

Before attempting to use the Plotter ROM and this manual you should be familiar with the operation of the HP-86/87 Personal Computer and especially with the graphics material presented in part 3 of the *HP-86/87 Operating and BASIC Programming Manual.* The *Plotter ROM Owner's Manual* discusses

only the graphics enhancements provided by the Plotter ROM and is not intended to be a complete guide to graphics.

Here is a list of the owner's documentation relevant to graphics programming.

1. Introductory manual for your computer: Tells you how to set-up your computer system (your computer plus any peripheral devices—printers, disc drive units, and plotters) and begin running BASIC programs. The set-up procedures include:
   - Initial set-up of the computer.
   - Setting the HP-IB interface select code.
   - Connecting peripheral devices and setting the device address.
2. *HP-86/87 Operating and BASIC Programming Manual:* A complete guide to HP enhanced BASIC, including all of your computer's integral statements and commands. The graphics statements discussed in this manual can be used to operate both the CRT and external plotters (with the addition of the Plotter ROM).
3. *HP 82936A ROM Drawer Instruction Sheet:* Tells you how to insert ROMs into the ROM drawer and install the drawer into the computer.
4. *Plotter ROM Owner's Manual:* Discusses the features of the Plotter ROM.
5. *HP-IB Interface Owner's Manual:* A complete guide to the HP-IB interface, including select code and device address information.
6. Peripheral device documentation: A guide to peripheral set-up and operation. For HP-GL devices this documentation includes a discussion of the HP-GL (Hewlett-Packard Graphics Language) instruction set.
7. *Graphics Pocket Guide:* A summary of all the graphics statements, functions, and error messages provided by your computer and the Plotter ROM.

Compatibility With Other Series 80 Products

The HP-86/87 Plotter ROM is compatible only with the HP-86/87 Personal Computer. Likewise, the HP-83/85 Plotter/Printer ROM is compatible only with the HP-83/85 Personal Computer. At power-on, the HP-86/87 searches for incompatible ROMs and returns a warning if an HP-83/85 ROM is present.

Syntax Guidelines

The syntax for ROM statements is presented in the same manner as syntax in your computer's operating and programming manual. The following conventions apply:

DOT MATRIX	Items in dot matrix are BASIC language keywords and punctuation that must be entered exactly as shown, except that lowercase letters can be substituted for uppercase letters.
[ ]	Brackets are used to enclose optional items.
italics	Items in italics are numeric and string expressions that must be included in the statement (unless enclosed by brackets).

ROM Installation

The Plotter ROM is added to your system in an HP 82936A ROM Drawer. Up to six different enhancement ROMs can be used in a single drawer. The procedure for installing ROMs and a ROM drawer is explained in the *HP 82936A ROM Drawer Instruction Sheet.* Make sure you *turn off the computer's power* whenever you add or remove ROMs and peripherals.

Memory Requirements

The Plotter ROM uses 1392 bytes of the computer's memory. Additional ROMs take up more memory. In the event that ROM memory requirements prevent you from loading a large program (the computer returns `ERROR 19 : MEM OVF`) and one or more of the HP 82907A, HP 82908A, or HP 82909A memory modules (32K, 64K, or 128K) or remove any unneeded ROMs from the ROM drawer.

Connecting Peripherals

Operation of peripheral devices such as plotters, printers, and disc drive units requires an interface device. The HP-87 comes equipped with an integrated HP-IB interface. Alternatively, you can use an accessory HP 82937A HP-IB Interface Module to connect HP-IB devices. Both are discussed in the *HP-IB Interface Owner's Manual.* If your plotting device uses an interface other than HP-IB, refer to the documentation accompanying the interface module for operating instructions.

Refer to the documentation accompanying your plotter for instructions about connecting a plotter to your computer and for special set-up procedures such as loading paper and changing pens. General instructions for connecting peripherals are also presented in section 3 of your computer's introductory manual.

Addressing the Plotting Device

The `PLOTTER IS` statement specifies the destination for all graphics output generated by your computer.

```
PLOTTER IS device selector
```

The device selector has the following form:

device selector = *interface select code* [*device address*]

The interface select code can be any number, variable, or expression between 1 and 10. A select code of 1 or 2 specifies the CRT as the graphics device. A select code between 3 and 10 routes the graphics output to the interface with that select code. The examples in this manual assume that the interface select code for the HP-IB interface is set to 7, the factory preset value.

With HP-IB devices, the two-digit decimal device address must be included in the PLOTTER IS device selector. Up to 14 devices can be connected to the HP-IB interface; the device address indicates which HP-IB peripheral is selected as the plotter.

All HP-IB devices have a group of address switches similar to the ones shown below. Refer to the documentation for your plotter for instructions about setting the device address. The device below is set to an address of 05, the factory preset value for all Hewlett-Packard plotters.



Examples:

PLOTTER IS 1 Selects the CRT as the graphics device.

PLOTTER IS 705 Selects the device with an address of 05 connected via an HP-IB interface with a select code of 7.

At power-on or whenever the computer is reset, the device selector defaults to PLOTTER IS 1 (selects the CRT).

In addition to selecting the plotting device, execution of the PLOTTER IS statement does the following:

- Activates the manually set graphics limits on the selected external plotter (refer to the graphics limits discussion below). PLOTTER IS 1 sets the graphics limits equal to the physical limits on the CRT.
- Sets the graphics default conditions. The graphics default conditions are also active at power-on or reset, whenever CRT memory is reapportioned, and whenever the LIMIT statement is executed. Refer to appendix A for a list of the graphics default conditions.

Setting the Graphics Limits

Graphics limits are the boundaries which enclose the rectangular plotting area on the CRT or the external plotter. The LIMIT statement specifies the graphics limits for the current plotting device. Alternatively, you can manually set the graphics limits on an external plotter (this feature is not available on the CRT) or use the default graphics limits.

Graphics limits are discussed briefly here as they pertain to external plotters. Refer to section 15 of your computer's operating and programming manual for a complete discussion of the LIMIT statement and other positioning and scaling operations. The LIMIT statement is also discussed in section 2 of this manual as it is used in digitizing the graphics limits.

The LIMIT Statement

The LIMIT statement specifies the graphics limits in millimeters for the current external plotter.

```
LIMIT [x min , x max , y min , y max]
```

The first two parameters specify the left and right limits and the second two parameters specify the lower and upper limits of the plotting area.

Example:

```
LIMIT 0,150,20,90
```

Specifies the graphics limits: left limit at 0 mm, right limit at 150 mm, lower limit at 20 mm, and upper limit at 90 mm.

LIMIT operates in the same manner on an external plotter as the CRT except:

- With an external plotter, LIMIT can be executed without parameters to digitize the graphics limits (refer to section 2).
- The allowable range of LIMIT parameters is larger for external plotters and varies for different devices (refer to the table below).
- An out-of-bounds LIMIT parameter is interpreted differently by external plotters and the CRT. If a LIMIT parameter is outside the physical limits of the CRT, the computer returns `Warning 101 : LIMIT OUT OF BOUND` and assigns the physical limit to the out-of-bounds parameter. If a LIMIT parameter is outside the physical limits of the selected external plotter, the computer returns `ERROR 89 : INVALID PARAM` and ignores the statement.

The allowable range of LIMIT parameters for the CRT graphics display and the HP 7470A Plotter are listed below. The LIMIT parameter range for the HP 7470A Plotter depends on the position of the A4/US switch located on the rear panel.

LIMIT **Parameter Range**

	x min	*x max*	*y min*	*y max*
CRT *Graph* Mode	0-125	0-125	0-75	0-75
CRT *Graph-all* Mode	0-171	0-171	0-75	0-75
HP 7470A Plotter (set to US)	0-257	0-257	0-191	0-191
HP 7470A Plotter (set to A4)	0-272	0-272	0-191	0-191

In addition to specifiying the graphics limits, the LIMIT statement activates the default graphics conditions (refer to appendix A).

Manually Set Limits

Hewlett-Packard plotters allow you to set the graphics limits manually, at the device. The procedure varies for different plotters; consult the documentation for your device for instructions about setting the limits manually. After entering the graphics limits manually, execute the PLOTTER IS statement. PLOTTER IS reads the manually set limits into computer memory.

Default Graphics Limits

At power-on, each external plotter has its own settings which are used to determine the graphics limits. When selected by the PLOTTER IS statement, the plotting device's default limits become active; these limits remain in effect until execution of a LIMIT statement, or until the limits are set manually and followed by execution of PLOTTER IS. The default limits for specific plotters are discussed in the documentation for the device.

> **Note:** To restore the default graphics limits at any time, turn the plotter off, then on, and execute the PLOTTER IS statement.

Interchanging CRT and Plotter Graphics

With minor modifications, many of your graphics programs written for the CRT will operate an external plotter. Differences between CRT and plotter graphics are summarized below. Before running your CRT graphics program on a plotter (or vice-versa), check for any graphics operations which are device-dependent.

- BPLOT and BREAD: The CRT byte plotting and byte reading operations are not possible with external plotters. These statements are ignored when the current plotting device is not the CRT.
- GRAPHALL: CRT memory reapportionment interferes with external plotting operations. Statements which reapportion CRT memory should precede any PLOTTER IS statement which selects an external device.
- DIGITIZE: You can digitize graphics with an external plotter; this capability is not provided for CRT graphics. The DIGITIZE statement is discussed in section 2.
- CURSOR: This statement is interpreted differently for CRT graphics and external plotter graphics. CURSOR is discussed in section 2.
- LABEL: The character set available for labeling with a plotter is not as extensive as for the CRT. The standard character set for external plotters is represented by the ASCII decimal code range 32 through 127; the CRT can label any character in the range 0 through 127. Refer to the documentation for your external plotter for treatment of characters 0 through 31.
- CSIZE: The default character size differs between external plotters and the CRT. The default values are CSIZE 5 for the CRT and CSIZE 3 for external plotters. Negative CSIZE height and aspect ratio parameters do not rotate and reflect labels on some external plotters. Refer to your computer's operating and programming manual for a complete discussion of the CSIZE statement.
- PEN: On multiple pen plotters, the PEN statement enables you to select different pen colors and widths for plotting. On the HP 7470A Plotter, PEN 1 selects the pen from the left stall, PEN 2 selects the pen from the right stall. PEN 0 returns the pen to the stall. Negative pen numbers are interpreted differently according to the capabilities of your plotter. If your CRT graphics program uses PEN -1 (plots black dots) or PEN -2 (plots exclusive or dots), change to a pen number suitable for your plotter. The default pen is PEN 1. This pen is selected whenever the default graphics conditions are in effect (for example after executing the PLOTTER IS or the LIMIT statement). With plotters that require manual pen changes, executing PEN 0 lifts the pen.
- GCLEAR: On plotters equipped with automatic paper advance, the GCLEAR statement advances the paper a full page. The HP 7470A Plotter and other manually loaded plotters ignore the GCLEAR statement.

- LINETYPE: Line type numbers 1 through 8 produce the same line type patterns for both the CRT and external plotters; the default length of the pattern varies (4 GUs for plotters, from 2 to 10 GUs for the CRT). Line type numbers outside of the range 1 through 8 are interpreted differently according to the capabilities of the plotter. Some plotters have more than eight available line types. If a plotter doesn't have the specified line type, it defaults to line type 1.

Example: The following is an example CRT graphics program from your computer's operating and programming manual. The program plots the function $y = \tan(x)/10$ on a labeled grid. The CRT graphics display was dumped (using the Plotter ROM) to an HP 82905B Printer.

```
10 ! *** TAN(X)/10  CRT ***
20 PLOTTER IS 1 ! ******** Specifies the CRT as the plotting device.
30 GRAPHALL !              Sets the display to graph-all mode.
40 LOCATE 20,190,10,90 !   Specifies the plotting boundaries.
50 SCALE -360,360,-1,1 !   Scales the plotting area.
60 DEG !                   Sets the computer to degrees mode.
70 FXD 0,2 !               Fixes 0 digits on x-axis labels, 2 on y-axis labels.
80 LGRID -10,.1,0,0,9,2 !  Draws a labeled grid.
90 MOVE -360,0 !           Moves the pen.
91 !
92 ! ******* Loop plots TAN(X)/10 *******
100 FOR X=-360 TO 360 STEP 5
110     IF X MOD 360=90 OR X MOD 360=270 THEN PEN UP @ GOTO ENDLOOP
120     PLOT X,TAN (X)/10
130 ENDLOOP: NEXT X !      End loop.
140 MOVE 0,1.1 !           Moves the pen.
150 LORG 4 !               Sets label origin 4.
160 CSIZE 6 !              Specifies character size.
170 LABEL "TAN(X)/10" !    Labels title.
180 END
```



To convert the example CRT graphics program to a plotter graphics program make the following changes:

1. Change the PLOTTER IS statement to the correct device selector (20 PLOTTER IS 1 → 20 PLOTTER IS 705).

2. Delete any CRT-specific operations that might affect external plotting operations. The `30 GRAPHALL` statement is deleted from the CRT graphics program since it interferes with external plotting operations.
3. Specify the graphics limits (either manually—at the plotter or by executing a `LIMIT` statement) to position and size the plot for the plotter (`30 LIMIT 0,250,20,140`).
4. Change the character size, if necessary, so the labels are sized correctly for the external plotter (`70 CSIZE 5`).
5. Add any desired enhancements that are available on your plotter such as pen color/width changes (`110 PEN 2`).
6. On multi-pen plotters end the program by storing the pen in its stall (`200 PEN 0`).

Here is the converted graphics program, adapted for the HP 7470A Plotter. The plotter output is shown on page 13.

```
10 ! *** TAN(X)/10  PLOTTER ***
20 PLOTTER IS 705 ! ****** Specifies the plotting device, select code = 7,
                           device address = 05.
30 LIMIT 0,250,20,140 !    Specifies the graphics limits in millimeters.
40 LOCATE 20,190,10,90 !   Specifies the plotting boundaries.
50 SCALE -360,360,-1,1 !   Scales the plotting area.
60 DEG !                   Sets the computer to degrees mode.
70 CSIZE 5 !               Specifies character size.
80 FXD 0,2 !               Fixes 0 digits on x-axis labels, 2 on y-axis labels.
90 LGRID -10,.1,0,0,9,2 !  Draws a labeled grid.
100 MOVE -360,0 !          Moves the pen.
101 !
102 ! ******* Loop plots TAN(X)/10 *******
110 PEN 2 !                Selects pen 2.
120 FOR X=-360 TO 360 STEP 5
130     IF X MOD 360=90 OR X MOD 360=270 THEN PEN UP @ GOTO ENDLOOP
140     PLOT X,TAN (X)/10
150 ENDLOOP: NEXT X !      End loop.
160 MOVE 0,1.1 !           Moves the pen.
170 LORG 4 !               Sets label origin 4.
180 CSIZE 6 !              Specifies character size.
190 PEN 1 @ LABEL "TAN(X)/10" !    Selects pen 1 and labels title.
200 PEN 0 !                Returns the pen to the stall.
210 END
```


TAN(X)/10
1.00
.80
.60
.40
.20
0.00
-.20
-.40
-.60
-.80
-1.00
-360
-270
-180
-90
0
90
180
270
360

Blank Page

Section 2

Digitizing Graphics

When equipped with the Plotter ROM, your computer is capable of digitizing graphics with a Hewlett-Packard external plotter. Digitizing is essentially the inverse of plotting. During *plotting* operations, the computer sends *x,y* coordinate values and pen status instructions (pen up or down) *to the plotter,* directing the pen to the specified location on the plotting area. A sequence of these instructions produces the points, lines, curves, and labels that comprise your graphics output. During *digitizing* operations, the plotter sends the current *x,y* coordinates and pen status information *to the computer.* Using the plotter's front panel controls, you can position the pen and change the up/down status manually, and digitize selected points from the plotting device.

The digitizing process enables you to convert graphics information into digital information. For example, you could trace the outline of a drawing or photograph with the plotter's pen or digitizing sight, digitizing the coordinates along the way. The pen coordinates and status are read into computer memory in a format identical to the PLOT statement. The graphics information can be digitized and then used with a PLOT statement to create a reproduction of the original graphics.

This section begins with a discussion of digitizing graphics limits and plotting boundaries. It is followed by a discussion of the DIGITIZE, CURSOR, and WHERE statements, which are used with an external plotter to digitize pen position and status.

Digitizing Limits and Boundaries

When executed without parameters, the LIMIT, LOCATE, and CLIP statements allow you to digitize the graphics limits or plotting boundaries on the external plotting device. This allows you to position and size your plot at the plotter instead of specifying the limit or boundary parameters in the LIMIT, LOCATE, and CLIP statements.

```
LIMIT
```

```
LOCATE
```

```
CLIP
```

Executing LIMIT without parameters suspends program execution. The computer waits to receive a message from the plotter containing the location of the lower-left and upper-right graphics limits.

Executing LOCATE or CLIP without parameters instructs the computer to wait and receive the location of the lower-left and upper-right CLIP or LOCATE plotting boundaries from the external plotter.

The procedure for digitizing the graphics limits or the plotting boundaries is:

1. Execute LIMIT, LOCATE, or CLIP; program execution is suspended.
2. Move the pen to the desired lower-left limit or boundary using the plotter's front panel controls and press the **ENTER** button. The pen position is sent to the computer where it is interpreted as the x min and y min limit or boundary parameters.
3. Move the pen to the desired upper-right limit or boundary and press the **ENTER** button again. The pen's position is sent to the computer and interpreted as the x max and y max limit or boundary parameters. The computer beeps each time the **ENTER** button is pressed to signify that the digitized information has been received.
4. The digitized graphics limits or plotting boundaries are now active; program execution continues.

Normally you would want to enter the lower-left limit or boundary first and the upper-right limit or boundary second. However, you can also digitize the graphics limits or LOCATE boundaries in different orientations to get a reflected image of your plot. For example, if you enter the upper-right limit first and the lower-left limit second, your plot will appear as if it was reflected through the origin. The procedure is analogous to exchanging parameters in the LIMIT or LOCATE statement. The sequence (first or second) and location (lower-left, upper-right, upper-left, or lower-right) of the digitized limit or boundary corner determines the type of reflection. The three types of reflections are summarized in the table below. Digitized CLIP boundaries cannot be used to reflect plots. Refer to your computer's operating and programming manual for more information about reflecting plots.

LIMIT and LOCATE Reflected Plots

	Unreflected plot	Reflection across origin	Reflection across x-axis	Reflection across y-axis
Location of first digitized limit or boundary	lower-left corner	upper-right corner	upper-left corner	lower-right corner
Location of second digitized limit or boundary	upper-right corner	lower-left corner	lower-right corner	upper-left corner

Example: The following program digitizes the graphics limits, frames the plotting area, and draws an arrow; the arrow points from the first digitized corner to the second digitized corner of the graphics limits. Experiment with your plotter by digitizing different graphics limits, and note how the shape and orientation of the figure changes. The example output below shows the unreflected plot and the three types of reflected images.

```
10 ! Digitize LIMITs
20 PLOTTER IS 705 !       Specifies the plotting device.
30 CLEAR @ DISP "DIGITIZE THE GRAPHICS LIMITS"
40 LIMIT !                Computer waits while you digitize the graphics limits
                          from the plotter.
50 CLEAR @ DISP "PLOTTING !"
60 FRAME !                Frames the digitized plotting area.
70 SCALE 0,10,0,10 !      Scales the digitized plotting area.
80 FOR I=1 TO 8 !         Loop plots figure.
90      READ X,Y
100     PLOT X,Y
110     PLOT 5,5
120     PLOT X,Y
130 NEXT I !              End loop.
140 DATA 8,8,1,5,3,4,1,2,2,1,4,3,5,1,8,8
150 PEN UP !              Lifts pen.
160 RESTORE @ GOTO 30 ! Restores DATA and repeats digitizing routine.
170 END
```



Digitizing Pen Location

We've already seen an example of digitizing, by executing the LIMIT statement without parameters and entering the limits from the plotter. This feature is just one application of your computer's digitizing capability. With the Plotter ROM and an external plotter, you can digitize any point on the plotting area and store it for later use.

Physical and Logical Pens

The ink pen on a pen plotter and the thermal print head on a thermal plotter are both considered "pens" in the sense that they draw the lines, points, and curves which constitute plotter graphics. Likewise, the CRT electron beam is considered a pen since it plots in an analogous manner, by exciting phosphorescent dots. Each of these types of pens which create the physical graphics image on the specific plotting device is referred to as the *physical pen.* On some devices, the physical pen location can be changed at the device, for example, by using the front panel pen movement controls on a pen plotter. The physical pen location can also be altered by executing a plotting statement (for example PLOT, MOVE, LABEL, or AXES). The physical pen is always located within the physical limits of the plotting device, but not necessarily within the current plotting area.

The computer has a pen of its own, the *logical pen.* The location and up/down status of the logical pen reside in computer memory and are determined by the most recently executed statement affecting pen location and status. For example, executing PLOT 10,20,1 lowers the logical pen at the coordinate position 10,20 (according to the current scaling units). The location and status of the logical pen are unaffected by the pen movement controls on the plotting device. The logical pen can be located anywhere inside or outside the physical limits of the device.

The physical and logical pens are recognized individually by the computer, but often coincide with each other during most plotting operations. Listed below, are some instances where the logical and physical pens take on different locations.

- Whenever the graphics default conditions are activated, the logical pen moves to the lower-left corner of the plotting area. The physical pen location is unaffected.
- When a plotting statement directs the pen to a point outside the current plotting area, the physical pen stops short of the intended point, at the current graphics limit or plotting boundary and is lifted (refer to the diagram below). In contrast, the logical pen location and status always coincide with the destination point and status specified by the plotting statement, regardless of whether or not the point lies within the current plotting area and whether or not it was actually plotted.
- Whenever the plotting device is changed, the physical and logical pens may have different locations depending on the initial physical pen position and the last executed plotting statement.
- Whenever the physical pen is moved using the pen movement controls at the external plotting device, the physical and logical pen locations differ.

The following diagram shows the location of the physical and logical pens during the sequence of plotting statements listed in the table below. The framed plotting area is scaled from 0 to 15 in the x direction and from 0 to 10 in the y direction. The solid black line indicates a line drawn during physical pen movement. The dashed black line indicates physical pen movement without line drawing. Note that the computer plots successive points according to the logical pen location.



Execute:	Resulting physical pen location and status	Resulting logical pen location and status
PLOT 10,8,1	(10,8) down	(10,8) down
DRAW 10,-5	(10,0) up	(10,−5) down
PLOT 5,5	(5,5) down	(5,5) down
MOVE -3,-3	(0,0) up	(−3,−3) up

Digitizing the Physical Pen Location

Digitizing the plotter's physical pen is an operation which involves both the plotter and the computer. The plotter's pen coordinates and up/down status are sent from the plotter to the computer and stored in three memory locations, identified by three numeric variables. Two variables identify the coordinate location of the physical pen; the third variable identifies the pen status.

When digitizing drawings, diagrams, photographs, or other types of graphics, it is the physical pen that you are digitizing. The physical pen digitizes and plots the physical images on the external plotter. There are two statements which can be used to digitize the physical pen: DIGITIZE and CURSOR. The syntax is identical for both statements, but the digitizing operations they perform are different.

DIGITIZE *x-variable* , *y-variable* [, *pen status variable*]

CURSOR *x-variable* , *y-variable* [, *pen status variable*]

DIGITIZE and CURSOR both assign the x-coordinate of the physical pen location to the x-variable parameter and the y-coordinate to the y-variable parameter. Pen coordinates are interpreted according to the current scale units. The optional third variable parameter is assigned the pen status information. If the pen is up, 0 is assigned to the variable. If the pen is down, 1 is assigned to the variable. All three parameters for DIGITIZE and CURSOR must be numeric variables.

The DIGITIZE and CURSOR statements use different methods for entering the digitized information.

- The DIGITIZE statement suspends program execution while you position the plotter's pen to the desired location and waits for you to press the **ENTER** button on the plotter. *The physical pen's coordinates and up/down status are read into computer memory only when the* **ENTER** *button is pressed.* When the computer receives the digitized information, program execution continues.
- The CURSOR statement does not suspend program execution. Thr physical pen's coordinates and up/down status are read into computer memory automatically without pressing the plotter's **ENTER** button.

Keep in mind that the pen must be positioned at the desired location for digitizing, prior to executing the CURSOR statement. The DIGITIZE statement allows you to position the pen and enter the digitized information after DIGITIZE is executed.

> **Note:** The DIGITIZE statement cannot be used for digitizing CRT graphics; if you attempt to DIGITIZE points on the CRT, the computer returns ERROR 109 : DIGITIZE. The CURSOR statement can be used to digitize CRT graphics, however there are no manual controls for altering the CRT's physical pen location.

Examples:

```
DIGITIZE XPOINT,YPOINT,STATUS
```

Suspends program execution. Pressing the **ENTER** button on the plotter assigns the x-coordinate location of the physical pen to the variable XPOINT, the y-coordinate location to the variable YPOINT, and the pen status to the variable STATUS. Program execution then continues.

```
CURSOR Xvalue,Yvalue
```

Assigns the x-coordinate of the physical pen location to the variable Xvalue, the y-coordinate to the variable Yvalue. Pen status information is not stored in computer memory, since the optional third variable is omitted.

Lift the plotter's pen and move it to any point using the pen movement controls. Lower the pen and execute:

```
CURSOR X,Y,P
DISP X,Y,P
```

The physical pen's coordinates and status are displayed on the CRT:

```
 53.6389633874          43.6944444444          1
```

Now, execute:

```
DIGITIZE xvar,yvar,pen
```

Lift the pen and relocate it on the plotter. Keep the pen up and press the plotter's **ENTER** button.

Execute:

```
DISP xvar,yvar,pen
```

The physical pen's new *x,y* coordinates and status are displayed on the CRT:

```
 31.9167109954          23.9027777778          0
```

Digitizing operations are commonly used for tracing drawings or other graphics images, which can then be reproduced using the PLOT statement. Note the similarity between the DIGITIZE statement and the PLOT statement.

DIGITIZE *x-variable* , *y-variable* [, *pen status variable*]

PLOT *x-coordinate* , *y-coordinate* [, *pen control*]

If the pen is in the appropriate up/down position while digitizing, the PLOT statement can use the pen status variable for the pen control parameter.

The digitized pen status variable takes on the value 0 or 1 depending on whether the pen is up or down. When the PLOT statement interprets a digitized pen status value as input for pen control, there are two possible results.

1. Pen status = 0 (for example, PLOT 4,-6,0): The pen is directed to the specified *x,y* coordinate and lifted. The pen maintains its initial up or down status until relocated at the specified *x,y* coordinate. If the pen is initially down, a line is drawn to the specified point and lifted. If the pen is initially up, the pen is moved to the specified point and remains up.
2. Pen status = 1 (for example, PLOT 10,14,1): The pen is directed to the specified *x,y* coordinate and lowered. The pen maintains it's initial up or down status until the pen is relocated.

The outcome of both PLOT statements is determined by the pen's up or down status preceding execution of PLOT. Therefore, when you digitize a point, have the pen set to the correct up/down status for the next digitized point.

Example: Digitize the simple drawing shown below. Keep in mind how pen status affects pen control when the digitized data is plotted using the PLOT statement.



The drawing consists of two line segments, requiring you to digitize four points (the endpoints of the two lines).

The following group of statements enables you to digitize four points on the currently selected external plotter:

```
FOR I=1 TO 4 @ DIGITIZE X(I),Y(I),P(I) @ NEXT I
```

The coordinate variables X(I),Y(I) are assigned the physical pen locations, according to the current scale. Pen status information is assigned to the variable P(I).

To digitize the example drawing, follow this sequence of steps:

1. Load a copy of the drawing, on paper, onto the plotter, and if available, install a digitizing sight; refer to the documentation accompanying your plotter for instructions.
2. Set the graphics limits manually, at the plotter, so that the drawing is located within the plotting area. Execute the PLOTTER IS statement to select the plotting device and activate the manually set limits.
3. Enter the above multi-line statement that digitizes four points, and press (END LINE).
4. Digitize the four points in the sequence shown on page 23, using the indicated up/down pen status. It is easiest to position the pen at the desired point while the pen or digitizing sight is down. When the pen is properly positioned and in the correct up/down status, press the plotter's **ENTER** button and move on to the next point.



The digitized coordinates and pen status information are stored in the numeric arrays X(I), Y(I), and P(I). To reproduce the digitized image, enter the following multi-statement line and press END LINE. Be sure your plotter is equipped with paper and a pen.

```
FOR I=1 TO 4 @ PLOT X(I),Y(I),P(I) @ NEXT I
```

If the physical pen coordinates and status were entered as shown, the plotter will duplicate the original drawing.

Keep in mind that the DIGITIZE and CURSOR statements digitize points according to the current scaling units. To recreate a digitized image accurately, the scaling units and plotting area dimensions in effect while plotting must match those in effect while digitizing.

Example: The following program digitizes up to 300 points on an external plotter and routes the digitized data to the current mass storage device. Use this program to digitize a drawing of the photograph shown below. After digitizing the final point, with the pen up, position the pen at the lower-left corner of the plotting area by pressing the **P1** or **LOWER LEFT** button on the plotter. Press the plotter's **ENTER** button to inform the computer that the pen is located at (0,0). The computer terminates the digitizing operation and closes the data file. If you're using a digitizing sight, install the sight after the plotter frames the digitized graphics limits.

```
10 ! *** DIGITIZE ***
20 PLOTTER IS 705 !                    Specifies the plotting device.
30 CLEAR @ DISP "ENTER DATA FILE NAME FOR DIGITIZE DATA"
40 INPUT F$ !                          Inputs string for data file name.
50 CREATE F$,300,24 !                  Creates a data file with 300 records,
                                       24 bytes each.
60 ASSIGN# 1 TO  F$ !                  Assigns buffer number 1 to data file F$.
70 CLEAR @ DISP "ENTER LIMITS FROM PLOTTER"
80 LIMIT @ FRAME !          Program execution is suspended; enter the graphics
                            limits. The plotting area is framed for reference.
90 OPTION BASE 1
100 DIM X(300),Y(300),P(300) !         Dimensions digitize array variables.
110 PN=0
120 PN=PN+1 !                          Increments digitize point number.
130 CLEAR @ DISP "POINT NUMBER";PN
140 DIGITIZE X(PN),Y(PN),P(PN) !       Digitizes pen position and status.
150 PRINT# 1,PN ; X(PN),Y(PN),P(PN) ! Prints digitized data to data file F$.
160 IF PN=300 THEN DISP "ARRAY IS FULL" @ GOTO 180 !  Halts digitizing operation
                                                      after 300 pts.
170 IF (X(PN) OR Y(PN))=1 THEN GOTO 120 ! Checks if pen is positioned at
171 !                                     lower-left corner of plotting area;
172 !                                     if not, digitizing continues.
180 ASSIGN# 1 TO  * !                  Closes the data file buffer number 1.
190 END
```



The following program plots the digitized data generated by the previous program. The program was repeated, using different digitized limits, to produce the output shown below.

```
10 ! *** PLOT DIGITIZED DATA ***
20 PLOTTER IS 705 !                    Specifies the plotting device.
30 CLEAR @ DISP "ENTER DIGITIZE DATA FILE NAME"
40 INPUT F$ !                          Inputs digitize data file name.
50 ASSIGN# 1 TO  F$ !                  Assigns buffer number 1 to data file F$.
60 OPTION BASE 1
70 DIM X(300),Y(300),P(300)
80 CLEAR @ DISP "ENTER LIMITS FROM PLOTTER"
90 LIMIT !                             Program execution is suspended while you
                                       digitize the limits.
100 I=0
110 I=I+1 !                            Increments digitize point number.
120 READ# 1,I ; X(I),Y(I),P(I)
130 PLOT X(I),Y(I),P(I)
140 IF (X(I) OR Y(I))=1 THEN GOTO 110 ! If x,y coordinates are not equal to 0,0
                                       the program continues to read and plot data.
150 ASSIGN# 1 TO  * !                  Closes the data file buffer number 1.
160 PEN 0 !                            Returns the pen to the stall.
170 END
```



Example: The following program enables you to trace a figure (up to 100 points) with the pen or digitizing sight, and plot the digitized image. Points are digitized using the CURSOR statement. Position the pen at the starting point of the figure to be traced, before you run the program. After you press (RUN), trace the drawing with the pen or digitizing sight using the plotter's front panel controls, as the CURSOR statement instructs the computer to read the pen location and status. To terminate the digitizing operation, lift the pen and move it to the lower-left corner of the plotting area by pressing the **P1** or **LOWER LEFT** button on the plotter.

```
10 ! *** CURSOR ***
20 PLOTTER IS 705 !                    Specifies the plotting device.
30 LIMIT 0,250,0,190 !                 Specifies the graphics limits.
40 OPTION BASE 1
50 DIM PTS(100,3) !                    Dimensions the digitize array variable.
60 PN=1
70 CURSOR PTS(1,1),PTS(1,2),PTS(1,3) ! Digitizes initial pen location, status.
80 PN=PN+1 !                           Increments point counter.
90 CLEAR @ DISP "POINT NUMBER";PN
100 CURSOR X,Y,P !                     Digitizes pen location and status.
101 ! ****** If the pen is moved 2 units or more, the digitized point is
    assigned to the array variable; the computer beeps to signify this outcome.
110       IF SQR ((X-PTS(PN-1,1))^2+(Y-PTS(PN-1,2))^2)<2 THEN 100
120       PTS(PN,1)=X @ PTS(PN,2)=Y @ PTS(PN,3)=P @ BEEP
121 ! ****** The digitizing operation continues until the array is full or the
    pen is located at the lower-left corner of the plotting area.
130       IF PN=100 THEN DISP "ARRAY IS FULL" @ GOTO 150
140       IF (PTS(PN,1) OR PTS(PN,2))=1 THEN 80
150 DISP "To plot the figure, remove the digitizing sight, load the paper and pr
ess [CONT]" @ PAUSE
160 PEN 1 !                                  Selects pen 1.
170 FOR I=1 TO PN
180 PLOT PTS(I,1),PTS(I,2),PTS(I,3) !        Plots the digitized data.
190 NEXT I
200 PEN 0 !                                  Returns the pen to the stall.
210 END
```

Digitizing the Logical Pen Location

The WHERE statement assigns the current *logical* pen coordinates and status to the specified variables. The parameters are the same as the parameters in the CURSOR and DIGITIZE statements.

WHERE *x variable , y variable* [*, pen status variable*]

The location and up/down status of the logical pen is determined by the most recently executed statement which changes pen status or location. All of the plotting statements which direct pen movement affect the logical pen location. In addition, statements and conditions which activate the graphics default conditions (refer to appendix A) lift the logical pen and move it to the origin (0,0). The physical pen's location and status are unaffected by the graphics default conditions.

The logical and physical pens often have the same location and status; any plotting statement which directs pen movement inside the current plotting area moves the *physical* pen as well as the *logical* pen.

Example: The following program demonstrates the difference between the *physical* and *logical* pen positions as read by the CURSOR and WHERE statements. When program execution is suspended, move the pen (using the plotter's front panel controls) to a new location, lower the pen, and press (CONT). The computer displays the resulting *physical* (CURSOR) and *logical* (WHERE) pen coordinate locations and status. In the example output below, the physical pen was moved to the coordinate location $x = 100.875140104$, $y = 72.3888888889$, and lowered.

```
10 ! *** WHERE ***
20 PLOTTER IS 705 !                    Specifies the plotting device.
30 MOVE 50,50 !                        Moves the pen to the point (50,50).
40 PAUSE !                             Pauses the program while you move the plotter
                                       pen to a new position.
50 WHERE WX,WY,WP !                    Assigns logical pen position and status to
                                       the variables WX,WY,WP.
60 CURSOR CX,CY,CP !                   Assigns physical pen position and status to
                                       the variables CX,CY,CP.
70 CLEAR
80 DISP "WHERE",WX,WY,WP !             Displays WHERE variables.
90 DISP "CURSOR",CX,CY,CP !            Displays CURSOR variables.
100 END
```

```
 WHERE                50                  50                  0
 CURSOR               100.875140104       72.3888888889       1
```

Example: The following program uses the WHERE statement to digitize the logical pen location as an aid in positioning labels in columnar form. The program prompts you to choose the number of sides for a regular polygon; the plotter labels the number of sides under a heading and draws the figure. The routine repeats five times, and places the successive side number labels in a column.

```
10  ! *** PLOT AND LABEL FIGURE ***
20 PLOTTER IS 705
30 LIMIT 50,200,0,190
40 SHOW 0,10,0,15
50 DEG @ COUNT=0 @ CSIZE 5,.5
60 MOVE 6,14 !                         Moves the pen to the column heading position.
70 LABEL "NUMBER OF SIDES" !           Labels the heading.
80 WHERE X,Y !                         Digitizes the pen location for the first column
                                       entry.
90 BEGIN: COUNT=COUNT+1
100 CLEAR @ DISP "CHOOSE AN INTEGER GREATER THAN OR EQUAL TO 3"
110 DISP "THE PLOTTER WILL DRAW A REGULAR POLYGON WITH THIS NUMBER OF SIDES"
120 INPUT SIDES
130 MOVE X,Y !                         Moves the pen to the digitized column position.

140 LABEL USING "8D" ; SIDES !         Labels the number of sides for the polygon.
150 WHERE X,Y !                        Digitizes the pen location for the next column
                                       entry.
160 MOVE 2,15-2.5*COUNT !              Positions the pen for plotting the polygon.
170 GOSUB PLOTFIG
180 IF COUNT<5 THEN GOTO BEGIN
190 PEN 0
200 END
201 !
202 !
203 !
210 PLOTFIG: FOR ANG=90 TO 450 STEP INT (360/SIDES)
220     RPLOT COS (ANG),SIN (ANG)
230 NEXT ANG
240 PEN UP
250 RETURN
260 END
```



NUMBER OF SIDES

3

4

5

6

12

Blank Page

Blank Page

Section 3

CRT Display Copying

The HP Plotter ROM enables you to copy, dot by dot, the contents of the alpha or graphics display to any of the following Hewlett-Packard graphics printers: HP 82905A, HP 82905B, HP 2631G, HP 2671G, and HP 2673A.

Some other HP graphics printers may also copy the display using the Plotter ROM. Refer to the documentation for your printer for information regarding its graphics capabilities.

Copying the Graphics Display

The DUMP GRAPHICS statement dumps the contents of the graphics display to the printer at the current PRINTER IS address. If the graphics display is in *graph* mode, the computer prints the *graph* mode display of 400 × 240 dots. If the CRT is in *graph-all* mode, the computer prints the 544 × 240 dot display.

DUMP GRAPHICS [*lower bound* , *upper bound* [, *rotate* [, *printer type*]]]

The optional lower and upper bound parameters are used to control what portion of the graphics display is dumped to the printer. Both are interpreted according to the current vertical scale. If for example, the current vertical scale is from 0 to 10, lower bound = 0 and upper bound = 5 indicates that the lower half of the display is to be dumped to the printer. The upper bound parameter must be greater than or equal to the lower bound, otherwise the computer returns Error 89 : INVALID PARAM. If omitted, the lower bound defaults to the bottom line of the display and the upper bound defaults to the top line, indicating that the entire display is to be dumped to the printer. The entire display is also dumped whenever the upper and lower bounds are equal. This allows you to specify the rotate and printer type parameters but still use the default upper and lower bounds.

The optional rotate parameter specifies whether a counterclockwise rotation of 90 degrees is performed on the output. Zero, the default value, signifies no rotation; any non-zero value rotates the output. When copying the *graph-all* mode display, some printers require that you rotate the output due to limitations in the number of available horizontal dot positions. If less than 544 dot positions are available, the right edge of the *graph-all* mode display will not be printed.

The last parameter specifies which type of graphics printer is to be used. It is interpreted in the following manner:

- Printer type = 0 (default value): indicates that the output is suitable for an HP 82905B Printer.
- Printer type < 0: indicates that the output is suitable for an HP 82905A Printer.
- Printer type > 0: indicates that the output is suitable for an HP standard graphics printer (HP 2631G, HP 2671G, HP 2673A).

If the wrong printer type is specified for your printer, the DUMP GRAPHICS output will be formatted incorrectly.

Examples:

`DUMP GRAPHICS`	All parameters omitted indicates that the entire display is to be dumped, without rotation, in the format required by an HP 82905B Printer.
`DUMP GRAPHICS 10,80,0,-1`	Dumps the graphics display from $y=10$ to $y=80$, without rotation, in the format required by an HP 82905A Printer.
`DUMP GRAPHICS 0,0,1,1`	Dumps the entire graphics display, rotated 90 degrees, in the format required by an HP standard graphics printer.

Note: In order to dump the graphics or alpha display to an HP standard graphics printer (HP 2631G, HP 2671G, HP 2673A), through an interface other than HP-IB, you need to remove the line feed character from the end-of-line sequence by executing `SET I/O 10,16,1`. (This assumes the select code for the printer interface is set to 10). Failure to remove double-spacing inserts blank lines between lines of output. Restore the line feed character for normal print operations by executing `SET I/O 10,16,2` after `DUMP GRAPHICS`. For more information regarding printer formatting, refer to your computer's operating and programming manual.

The size and aspect ratio of the `DUMP GRAPHICS` output varies for different printers, and is different than the CRT graphics display due to differences in raster spacing among devices.

Example: The following program draws the curve $y=\sin(x)/x$. Plot the function and execute the `DUMP GRAPHICS` statement. Be sure to set the printer address (`PRINTER IS`) and suppress the double-spacing (`SET I/O`) if necessary.

```
10 ! *** sin(x)/x ***
20 PLOTTER IS 1
30 GCLEAR
40 RAD @ SCALE -(4*PI ),4*PI ,-.5,1
50 FOR ANG=-(4*PI ) TO 4*PI  STEP PI /50
60     PLOT ANG,SIN (ANG)/ANG
70     PLOT ANG,0
80 NEXT ANG
90 GRID PI /4,.1
100 END
```

Here are some examples of printer output resulting from the corresponding `DUMP GRAPHICS` statements. The graphics was dumped to an HP 82905B Printer (the default printer type).

```
DUMP GRAPHICS
```



```
DUMP GRAPHICS 0,1
```



```
DUMP GRAPHICS 0,.5,1
```



Copying the Alpha Display

The DUMP ALPHA statement dumps the visible contents of the alpha display to the printer at the current PRINTER IS address. The alpha display window contains either 16 (PAGESIZE 16) or 24 (PAGESIZE 24) lines. The page is rotated 90 degrees counterclockwise on the printer output. Rotation is implemented automatically in order to fit the 544 dot wide alpha display onto printers which cannot accomodate 544 dots in the horizontal direction.

DUMP ALPHA [*printer type* [, *cursor on*]]

The optional printer type parameter specifies for which type of device the output is formatted, and is interpreted the same as for the DUMP GRAPHICS statement. The default printer type is 0, indicating an HP 82905B Printer is to be used.

The optional cursor on parameter signifies whether the cursor is to be printed or not. If omitted, or equal to zero, the cursor is not printed. Any non-zero value causes the cursor to be printed.

Examples:

DUMP ALPHA	All parameters omitted indicates that the alpha display is dumped, without the cursor, in the format required by an HP 82905B Printer.
DUMP ALPHA 1,1	Dumps the alpha display, with the cursor, in the format required by an HP standard printer.

When dumping the alpha display to an HP standard graphics printer (HP 2631G, HP 2671G, HP 2673A) connected via a non-HP-IB interface, use the SET I/O statement, as described on page 32, to suppress the automatic line feed. Otherwise, the DUMP ALPHA output will contain blank lines.

As with DUMP GRAPHICS, the size and aspect ratio of the DUMP ALPHA output varies for different printers, and is different than the CRT alpha display due to varying raster spacing among devices.

Example: The following program produces a box within the alpha display and dumps the display to an HP 82905B Printer (the default printer type). The resulting output is shown on page 36.

```
10 ! *** DUMP ALPHA ***
20 PRINTER IS 701 !                          Specifies the printer address.
30 CLEAR !                                   Clears the alpha display.
40 DIM BAR$[80]
50 BAR$="" @ C$=CHR$ (160)
60 FOR I=1 TO 80
70     BAR$=BAR$&C$
80 NEXT I
90 DISP BAR$
100 FOR ROW=1 TO 11
110     IF ROW=6 THEN DISP C$;TAB (33);"ALPHA DISPLAY";TAB (80);C$ ELSE DISP C$;T
AB (80);C$
120 NEXT ROW
130 DISP BAR$
140 DUMP ALPHA !                             Dumps the alpha display to the
                                             printer, without the cursor.
150 END
```


ALPHA DISPLAY

Blank Page

Blank Page

Section 4

Maintenance and Service

Maintenance

The Plotter ROM does not require maintenance. However, there are several areas of caution that you should note. They are:

WARNING: Do not place fingers, tools, or other foreign objects into the plug-in ports. Such actions may result in minor electric shock hazard and interference with some pacemaker devices. Damage to plug-in port contacts and the computer's internal circuitry may also result.

CAUTION: Always switch off the computer and any peripherals involved when inserting or removing modules. Use only plug-in modules designed by Hewlett-Packard specifically for your computer. Failure to do so could damage the module, the computer, or the peripherals.

CAUTION: If a module or ROM drawer jams when inserted into a port, it may be upside down or designed for another computer. Attempting to force it may damage the computer or the module. Remove the module carefully and reinsert it.

CAUTION: Do not touch the spring-finger connectors in the ROM drawer with your fingers or other foreign objects. Static discharge could damage electrical components.

CAUTION: Handle the plug-in ROMs very carefully while they are out of the ROM drawer. Do not insert any objects in the contact holes on the ROM. Always keep the protective cap in place over the ROM contacts while the ROM is not plugged into the ROM drawer. Failure to observe these cautions may result in damage to the ROM or the ROM drawer.

For instructions on how to insert and remove the ROM and the ROM drawer, please refer to the ROM drawer instruction sheet or section 2 of your computer's introductory manual.

Service

If at any time you suspect that the ROM drawer or the Plotter ROM may be malfunctioning, do the following:

1. Turn the computer and all peripherals off. Disconnect all peripherals and remove the ROM drawer from the port. Turn the computer back on. If it does not respond or displays `Error 23 : SELF TEST`, the computer requires service.
2. Turn the computer off. Install the ROM drawer, with the Plotter ROM installed, into any port. Turn the computer back on.
   - If the cursor does not appear, the system is not operating properly. To help determine what is causing the improper operation, repeat step 2 with the ROM drawer installed in a different port, both with the Plotter ROM installed in the ROM drawer and with the Plotter ROM removed from the ROM drawer.

- If `Error 112 : PLOTTER ROM` is displayed, indicating that the ROM is not operating properly, turn the computer off and try a different ROM drawer slot. This will help you determine if particular slots in the ROM drawer are malfunctioning, or if the ROM itself is malfunctioning.

3. Refer to Obtaining Repair Service for information on repair service for the malfunctioning device.

Warranty Information

The complete warranty statement is included in the information packet shipped with the ROM. Please retain this statement for your records.

If you have any questions concerning this warranty, please contact:

In the U.S.: One of the six Field Repair Centers listed on the Service Information Sheet packaged with your owner's documentation.

In other countries: Contact your nearest sales and service facility. If you are unable to contact that facility, please contact:

In Europe:

Hewlett-Packard
7, rue du Bois-du-Lan
P. O. Box
CH-1217 Meyrin 2
Geneva
Switzerland
Tel. (22) 82 70 00

Other countries:

Hewlett-Packard Intercontinental
3495 Deer Creek Rd.
Palo Alto, California 94304
U.S.A.
Tel. (415) 857-1501

Obtaining Repair Service

Not all Hewlett-Packard facilities offer service for your computer. For information on obtaining service in your area, consult the service information included in the Service Information Sheet packaged with your computer, or contact your authorized HP Series 80 dealer or the nearest Hewlett-Packard sales and service facility.

If your computer requires repair, you can help assure efficient servicing by following these guidelines:

1. Leave the configuration of the computer exactly as it was at the time of the malfunction; any plug-in modules and flexible discs in use at that time should be kept in place.
2. Write a description of the malfunction symptoms for service personnel.
3. Save printouts or any other materials that illustrate the problem.
4. Have on hand a sales slip or other proof of purchase to establish warranty coverage period.

Serial Number

Each Series 80 computer carries an individual serial number plate on the rear panel. We recommend that owners keep a separate record of this number. Should your unit be lost or stolen, the serial number is often necessary for tracing and recovery, and for any insurance claims.

Hewlett-Packard does not maintain records of individual owners of Series 80 computers and unit serial numbers.

General Shipping Instructions

Should you ever need to ship your computer, be sure it is packed in a protective package to avoid in-transit damage. Use the original shipping case if possible. Shipping damage is not covered by the warranty. All customs and duties are the customer's responsibility.

Hewlett-Packard recommends that the customer always insure shipments.

Potential for Radio/Television Interference

The Plotter ROM uses radio frequency energy and may cause interference to radio and television reception. The ROM has been type-tested and found to comply with the limits for a Class B computing device in accordance with the specifications in Subpart J of Part 15 of the FCC Rules. These specifications provide reasonable protection against such interference in a residential installation. However, there is no guarantee that interference will not occur in a particular installation. If the ROM does cause interference to radio or television, which can be determined by turning the computer on and off with the ROM installed and with the ROM removed, you can try to eliminate the interference problem by doing one or more of the following:

- Reorient the receiving antenna.
- Change the position of the computer with respect to the receiver.
- Move the computer away from the receiver.
- Plug the computer into a different outlet so that the computer and the receiver are on different branch circuits.

If necessary, consult an authorized HP dealer or an experienced radio/television technician for additional suggestions. You may find the following booklet, prepared by the Federal Communications Commission, helpful: *How to Identify and Resolve Radio-TV Interference Problems.* This booklet is available from the U.S. Government Printing Office, Washington, D.C. 20402, Stock No. 004-000-00345-4.

Blank Page

Appendix A

Graphics Default Conditions

The graphics default conditions are active whenever:

- The computer is turned on or reset by pressing (RESET).
- The PLOTTER IS statement or the LIMIT statement is executed.
- CRT memory is reapportioned.

The default conditions are as follows:

1. Plotting boundaries (set by CLIP and LOCATE) are set to the graphics limits.
2. The plotting area is scaled in graphics units (GUs), the default scale.
3. The computer is set to user units mode with user units (UUs) equal to graphics units (GUs).
4. Pen color is set to PEN 1.
5. Lines are drawn in solid line type (LINE TYPE 1).
6. Labels are drawn using the standard character size (CSIZE 5 for the CRT, CSIZE 3 for external plotters).
7. Labels are positioned according to label origin 1 (LORG 1).
8. Labeling direction is left-to-right (LDIR 0).
9. The logical pen moves to the origin (lower-left corner).

Each of the terms and statements referred to above are discussed in detail in your computer's operating and programming manual.

Blank Page

HP-GL Instructions

The PRINTER IS and PRINT statements are used to send HP-GL instructions and control code sequences to an external plotter. HP-GL is Hewlett-Packard Graphics Language, the internal language of Hewlett-Packard plotters. A discussion of the HP-GL instruction set is included in the documentation for your plotter. In most cases, you will want to use the graphics statements discussed in your computer's operating and programming manual and in this manual. The HP-GL instructions can be used to access any special capabilities of your external plotter.

Example: The following program demonstrates some features of the HP-GL instruction set for the HP 7470A Plotter. By specifying the plotter's device selector in the PRINTER IS statement, and sending the appropriate HP-GL instruction with a PRINT or PRINT USING statement, the plotter can be used as the system printer to label program listings. You can also use an HP-GL instruction to slow the pen speed.

```
10 ! *** HP-GL ***
20 PRINTER IS 705,120 !     Specifies the plotter as the printer with 120 charact
ers per line.
30 PRINT "SI.15,.4" !       Specifies the absolute character size in centimeters:
 width=.15 cm, height=.4 cm.
40 PRINT "IW;SP1;PA800,6000" ! Specifies plotting window, pen 1, and moves pen.
50 PRINT USING "AA" ; "LB" !  Sets the plotter to label mode.
60 PLIST 10,30 !            Lists statements 10 through 30.
70 PRINT "ñ" ! Takes plotter out of label mode (press [CTRL] C to produce ñ).
80 PRINT "VS 2" !           Specifies approximate pen velocity = 2 cm/second.
90 PRINT USING "AA" ; "LB" ! Sets the plotter to label mode.
100 PLIST 80,80 !           Lists statement 80.
110 PRINT "ñ" !             Takes plotter out of label mode.
120 END
```

```
10 ! *** HP-GL ***
20 PRINTER IS 705,120 !     Specifies the plotter as the printer with 120 characters per line.
30 PRINT "SI.15,.4" !       Specifies the absolute character size in centimeters: width=.15 cm, height=.4 cm.

80 PRINT "VS 2" !           Specifies approximate pen velocity = 2 cm/second.
```

Blank Page

Plotter ROM Syntax Summary

Syntax Guidelines

DOT MATRIX	Items in dot matrix must be entered exactly as shown, except that lowercase letters can be substituted for uppercase letters.
[ ]	Parameters enclosed in brackets are optional items.
italics	Items in italics are numeric and string expressions that must be included in the statement (unless enclosed by brackets).

Plotter ROM Statements

The following list of graphics statements includes those which are integral to the Plotter ROM, and any statements integral to the computer which take on special functions when used with an external plotter. For a complete list of all the graphics statements for both your computer and the Plotter ROM, refer to the *Graphics Pocket Guide.*

CLIP [*x min* , *x max* , *y min* , *y max*] **Page 15**

CURSOR *x-variable* , *y-variable* [, *pen status variable*] **Page 19**

Pen up: pen status equals 0.
Pen down: pen status equals 1.

DIGITIZE *x-variable* , *y-variable* [, *pen status variable*] **Page 19**

DUMP ALPHA [*printer type* [, *cursor on*]] **Page 34**

Printer type = 0: output is formatted for an HP 82905B Printer.
Printer type < 0: output is formatted for an HP 82905A Printer.
Printer type > 0: output is formatted for an HP 2631G, HP 2671G, or HP 2673A Printer.

DUMP GRAPHICS [*lower bound* , *upper bound* [, *rotate* [, *printer type*]]] **Page 31**

LIMIT [*x min* , *x max* , *y min* , *y max*] **Page 9**

LOCATE [*x min* , *x max* , *y min* , *y max*] **Page 15**

PLOTTER IS *device selector* **Page 7**

device selector = *interface select code* [*device address*]

WHERE *x-variable* , *y-variable* [, *pen status variable*] **Page 26**

Blank Page

Plotter ROM Error Messages

A complete list of your computer's error messages is presented in appendix F of its operating and programming manual. In addition to these errors, there are two errors that are specific to the Plotter ROM. Plotter ROM errors are identified by ERROM number 240.

Error Message	Error Condition
112 : PLOTTER ROM	The Plotter ROM failed self-test. If this error should occur, do not use the ROM. Contact your local HP Series 80 dealer or the nearest Hewlett-Packard sales and service facility.
126 : PLOTTER	The addressed peripheral device does not respond within 2 seconds after it is addressed with a PLOTTER IS statement.

Personal Computer Division
1010 N.E. Circle Blvd., Corvallis, OR 97330, U.S.A.

Reorder Number
00087-90109

Printed in U.S.A. 6/82
00087-90285